
Week 6: Curves, Surfaces and Maple

In this document, we take a look at defining curves and surfaces, both in Calculus and in
Maple. We’ll also look at some commands used to analyze these- limits and derivatives.

Curves

During our first lab, we talked about curves (or paths) in the plane. We will see that
there are lots of ways of defining a curve in the plane (or in three dimensions). Here, we
look at a few:

As the Graph of y = f(x)

We can define a curve in the plane as being the graph of y = f(x). That is, one way of
defining a curve C is formally:

C = {(x, y) | y = f(x), a ≤ x ≤ b}

We have seen how to plot these kinds of curves in Maple. For example (see Figure 1 below),
here we plot an expression F:

F:=cos(x/2)+sin(2*x);

plot(F, x=0..7);

As a Set of Parametric Functions

If the curve is NOT a function of x (or y for that matter), often we can make our curve a
function by finding parametric equations for x and y in terms of a third variable- usually
denoted by t.

Using the notation we typically see in a Calc III course, that would be (either in two
dimensions or three dimensions):

〈x(t), y(t)〉, a ≤ t ≤ b or 〈x(t), y(t), z(t)〉, a ≤ t ≤ b

The difference between these can also be clearly demonstrated graphically. In Figure 1 below,
we graph on the left represents y as a function of x. On the right, both x and y are functions
of t.

When plotting parametric curves in Maple, we should remember that Maple distinguishes
between expressions and functions. We have always plotted using expressions, so we’ll show
that again here. To get something similar to Figure 1, we would type:

F:=sin(2*t); G:=cos(sqrt(2)*t);

plot( [F,G,t=0..20]);

To plot two functions together using t as the horizontal axis, we would type:

Figure 1: Curves in the plane. On the left, y = f(x). On the right, the curve has been
parameterized in t. That is, x(t) = sin(2t) and y(t) = cos(

√
2 t). The curve on the right

cannot be expressed as y = f(x), and is a more general way of defining a curve (or path).

plot([F,G],t=0..20);

If we want two parametric sets of functions, for example, the previous graph together with
the unit circle, we might write it as:

plot([ [F,G,t=0..20],[cos(t), sin(t),t=0..2*Pi] ]);

A Third Way: Implicitly Defined Curves

In Calculus II and III, we have treated expressions like: f(x, y) = k (with k a constant) as
defining y in terms of x implicitly. For example,

cos(xy) =
1

3

gives the set of points in the plane shown in Maple by:

with(plots):

implicitplot(cos(x*y)=1/3,x=-4..4,y=-4..4);

If you run this command, you may notice that some of the curves are not smooth. We can
fix this by using a finer grid of points for the plot:

implicitplot(cos(x*y)=1/3,x=-4..4,y=-4..4, grid=[90,90]);

What you are looking at can be viewed as the level curve (or contour) of the surface z =
cos(xy) at z = 1/3. To visualize this in Maple, we can type:

contourplot3d(cos(x*y),x=-4..4,y=-4..4,filledregions=true);

Figure 2: An implicitly defined function, cos(xy) = 1/3, and the contours of z = cos(xy).

Curves in 3-d: “Spacecurves”

We can parameterize a curve in three dimensions simply by adding a third coordinate that
depends on t. That is, a curve C in 3-d is defined as:

〈x(t), y(t), z(t)〉 a ≤ t ≤ b

Maple needs a new command for curves in three dimensions- It refers to them as space
curves. Here’s an example, where we plot a curve called a knot. In particular, notice that
we need with(plots) at the top of the worksheet.

with(plots):

F:=-10*cos(t)-2*cos(5*t)+15*sin(2*t);

G:=-15*cos(2*t)+10*sin(t)-2*sin(5*t);

H:=10*cos(3*t);

spacecurve( [F,G,H, t=0..2*Pi] );

Surfaces

We usually start discussing surfaces in Calculus III using the graph of z = f(x, y). That is,
our z value is a function of (x, y).

As with curves, this is very restrictive in the sense that there are many surfaces that are

not strictly functions of x, y- For example, the surface of a sphere.

x2 + y2 + z2 = 1 ⇒ z = ±
√

1− x2 − y2

A surface more generally can be thought of (parametrically) as the image of a mapping of
the plane into three dimensions. That is, x, y, and z can each be a function of two variables
(u, v).

In brief, that means that a surface is the graph (in 3-d) of the set of parametric functions:

〈x(u, v), y(u, v), z(u, v)〉, a ≤ u ≤ b, c ≤ v ≤ d

Surfaces as z = f(x, y)

For surfaces as the graph of z = f(x, y), we can parametrize easily:

x = u
y = v
z = f(u, v)

⇒ 〈u, v, f(u, v)〉

In Maple, we can define a surface either as z = f(x, y) or parametrically. The following two
plots should show the same thing:

with(plots):

plot3d(sqrt(1-x^2-y^2),x=-1.5..1.5,y=-1.5..1.5,

axes=NORMAL, scaling=CONSTRAINED);

plot3d([u,v,1-u^2-v^2,u=-1.5..1.5,y=-1.5..1.5,

axes=NORMAL, scaling=CONSTRAINED);

You can also plot the surface of a sphere parametrically, which you may not have seen
before. We define the functions for x, y and z as F,G and H to get the upper half sphere:

F:=cos(u)*sin(v);

G:=sin(u)*sin(v);

H:=cos(v);

plot3d([F,G,H],u=0..2*Pi, v=0..Pi/2, scaling=CONSTRAINED);

Example: Surface of Revolution

Let’s try an example: Plot the surface obtained by rotating the graph of y = 1+x2, x = 0..2
about the x−axis.

SOLUTION: Take any point on the curve, and imagine what happens as this point
revolves about the x−axis- We get a circle... Putting the z−axis so that it points at you,
that circle will be parallel with the (y, z) plane, and suppose that the circle has radius R.

Then, as we have learned in the past, we can parametrize y and z as:

y = R cos(v) z = R sin(v)

Finally, R is the length from the x−axis to the curve, which is f(x). Therefore, we have:

〈x, (1 + x2) cos(v), (1 + x2) sin(v)〉, 0 ≤ x ≤ 2, 0 ≤ v ≤ 2π

In Maple, we would type the following (you do not need the line breaks in Maple- These are
just for readability). By the way, you might see what it really looks like by constraining the
scaling.

plot3d([x,(1+x^2)*cos(v),

(1+x^2)*sin(v)], x=0..2,

v=0..2*Pi);

Example: Surface of Revolution 2

Let’s try an example: Plot the surface obtained by rotating the graph of y = x2, x = 0..1
about the y−axis.

SOLUTION: Take any point on the curve, and imagine what happens as this point
revolves about the y−axis- We get a circle... Putting the z−axis so that it points at you,
that circle will be parallel with the (x, z) plane, and suppose that the circle has radius R.

Then, as we have learned in the past, we can parametrize y and z as:

x = R cos(v) z = R sin(v)

Finally, R is the length from the y−axis to the curve, which is f(y). Therefore, we have:

〈√y cos(v), y,
√
y sin(v)〉, 0 ≤ y ≤ 1, 0 ≤ v ≤ 2π

In Maple:

plot3d([ sqrt(y)*cos(v), y,

sqrt(y)*sin(v)], y=0..1,

v=0..2*Pi);

5



Example 3: Color and Animation Examples

• Plot the surface z = xe−x2−y2 , and color the graph according to sin(xy):

plot3d(x*exp(-x^2-y^2), x = -2..2, y =-2..2, color = sin(x*y))

• Here is an example of an animation, where we vary the height of the valley and hilltop
by using an extra parameter A:

animate( plot3d, [sin(A)*x*exp(-x^2-y^2), x=-2..2, y=-2..2],

A=0..2*Pi,shading=zhue );

– Click on the graph and there will be a play button on the menu. There is also an
option to run a loop- See if you can find it.

– You can save your animation as an animated GIF file- Right-click the graph, then
Export..., and choose GIF. To see the animated GIF file, you can open it using
Chrome or Firefox.

Example 4: Multiple Surfaces

Just as with curves, we can plot multiple surfaces together. Two ways are shown below:

Method 1: Plot the two separately, then “display” the results. Use this technique if you
want to overlay graphs over the top of each other (or with each other).

For example:

with(plots):

A:=plot3d(sin(x)*sin(y),x=-Pi..Pi,y=-Pi..Pi,color=white):

B:=plot3d(x*exp(-x^2-y^2), x = -2..2, y =-2..2, color = sin(x*y));

display({A,B});

Method 2: If your functions are all of the form z = f(x, y), you can use a single plot
command. For example, here we plot three surfaces: z = sin(xy), z = x−y and z = 4e−x2−y2 ,
all on the same graph:

plot3d({sin(x*y), x-y, 4*exp(-x^2-y^2)},x=-2..2,y=-2..2);

Question: What happens if you use square brackets around the functions instead of curly
braces?... Does the same thing happen with 4 functions?

6


